




Game - Electronic Dice with Last Number Recall


Block Diagram:



APPLICATION OPERATION:

When you press the switch for a short time, a random number is displayed for a few seconds. If you press the switch for a long time (about 3 secs) the last number is blinking while the switch remains pressed. When not in use, the PICmicro™ is in sleep mode.

Graphical hardware representation:



Microchip Technology Incorporated, has been granted a nonexclusive, worldwide license to reproduce, publish and distribute all submitted materials, in either original or edited form. The author has affirmed that this work is an original, unpublished work and that he/she owns all rights to such work. All property rights, such as patents, copyrights and trademarks remain with author.

Automate the Home

APPENDIX A: SOURCE CODE

```
; EDice - Electronic Dice
; Philippe Labonne (PLAB97)
; (819) 538-2169
; 5450 18 Street
; Grand-Mere, Quebec
; G9T-6P1, Canada

LIST      P=12C508                ; Select PIC12C508

INDF      EQU      H'0000'        ; Init.
TMR0      EQU      H'0001'
PCL       EQU      H'0002'
STATUS    EQU      H'0003'
Z         EQU      H'0002'        ; Adr. of Z flag in Status
FSR       EQU      H'0004'
OSCCAL    EQU      H'0005'
GPIO      EQU      H'0006'
DICENUM   EQU      H'0007'        ; Dice Number
TMR1      EQU      H'0008'        ; First timer
TMR2      EQU      H'0009'        ; Second timer
ROLL      EQU      H'000A'        ; Roll the dice ($0 to $5)
COUNTER   EQU      H'000B'        ; First Switch counter
COUNTER1  EQU      H'000C'        ; Second switch counter

        MOVLW      0x028          ; I/O Direction
        TRIS       GPIO           ; IOI000
        MOVLW      0x5F           ; Option register config
        OPTION     ; 01011111
        MOVLW      17h            ; Start with all leds off
        MOVWF      GPIO           ; 010111
        CLRF       COUNTER

; *****
MAIN    CALL      ROLLIT
        BTFSC      GPIO,3         ; Look if switch is pressed
        GOTO       SWPRESS        ; yes goto swpress
        MOVLW      0x02           ; No, look if last switch was good
        SUBWF      COUNTER,0      ;
        BTFSC      STATUS,0       ;
        GOTO       GOODNUM        ; Yes, display a new number
        CLRF       COUNTER        ; else, clear the counter
        MOVLW      0xFF           ;
        TRIS       GPIO           ; Set all IO in input mode
        MOVF       GPIO          ;
        SLEEP      ; Enter sleep mode
; *****
SWPRESS INCF       COUNTER1        ; Inc second counter
        BTFSC      STATUS,Z       ; If counter roll over, add 1 to first cnt
        INCF       COUNTER        ; Inc counter
        MOVLW      0xFF           ; Check if counter = FF
        SUBWF      COUNTER,0      ;
        BTFSC      STATUS,Z       ;
        GOTO       SHOWLAST       ; If yes Show last number
        GOTO       MAIN          ; Else go back to main
; *****
ROLLIT  MOVLW      0x05           ; Look if number is 6 ($05)
        SUBWF      ROLL,0         ;
        BTFSC      STATUS,Z       ;
        GOTO       SETONE         ; Yes, set number 1 ($0)
        INCF       ROLL           ; Else, inc roll
        RETLW      0x00           ; Return
SETONE  MOVLW      0x00           ; Set roll = 1 ($0)
        MOVWF      ROLL           ;
        RETLW      0x00           ; Return
; *****
GOODNUM CALL      TABLE          ; Call table to translate number
```

Automate the Home

```
        MOVWF    DICENUM        ; Save value
        MOVWF    GPIO           ; Display value
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CALL     DELAY           ;
        CLRF     COUNTER         ; Clear counter
        MOVLW    0x17            ; Put all leds off
        MOVWF    GPIO           ;
        GOTO     MAIN            ; Back to main routine
; *****
SHOWLAST MOVF     DICENUM,0      ; Load last number
        MOVWF    GPIO           ; Display number
        CALL     DELAY           ; Call delay
        MOVLW    0x17            ; Put all leds off
        MOVWF    GPIO           ;
        CALL     DELAY           ; Call delay
        BTFSC    GPIO,3         ; look if switch is always pressed
        GOTO     SHOWLAST       ; yes show last again (blinking)
        CLRF     COUNTER         ; clr counter
        MOVLW    0x17            ; Else, put all leds off
        MOVWF    GPIO           ;
        GOTO     MAIN            ; Return to main routine
; *****
DELAY     MOVLW    0xFF          ; Set tmr2 to $FF
        MOVWF    TMR2           ;
INDEL1    MOVLW    0xFF          ; Set Tmr1 to FF
        MOVWF    TMR1           ;
INDEL2    DECFSZ   TMR1,1        ; Dec tmr1
        GOTO     INDEL2         ; goto indel2
        DECFSZ   TMR2,1        ; Dec tmr2
        GOTO     INDEL1         ; goto indel1
        RETLW    0x00          ; return
; *****
TABLE     MOVLW    0x00          ; Look if number is 1 ($00)
        SUBWF    ROLL,0         ;
        BTFSC    STATUS,Z       ;
        RETLW    0x13          ; Yes, return with ($13)
        MOVLW    0x01          ; Look if number is 2 ($01)
        SUBWF    ROLL,0         ;
        BTFSC    STATUS,Z       ;
        RETLW    0x07          ; Yes, return with ($07)
        MOVLW    0x02          ; Look if number is 3 ($02)
        SUBWF    ROLL,0         ;
        BTFSC    STATUS,Z       ;
        RETLW    0x03          ; Yes, return with ($03)
        MOVLW    0x03          ; Look if number is 4 ($03)
        SUBWF    ROLL,0         ;
        BTFSC    STATUS,Z       ;
        RETLW    0x06          ; Yes, return with ($06)
        MOVLW    0x04          ; Look if number is 5 ($04)
        SUBWF    ROLL,0         ;
        BTFSC    STATUS,Z       ;
        RETLW    0x02          ; Yes, return with ($02)
        MOVLW    0x05          ; Look if number is 6 ($05)
        SUBWF    ROLL,0         ;
        BTFSC    STATUS,Z       ;
        RETLW    0x04          ; Yes, return with ($04)

        END                    ; Will never pass here ;-)
```

Automate the Home

NOTES: